Câble Coaxial

Banque PT 2008 A Première partie : paramètres primaires d'une ligne coaxiale

Nous donnons simplement des réponses en faisant usage de grandeurs réduites.
Une grandeur réduite est une mesure (un nombre pur) d'une grandeur physique dimensionnée pour laquelle on a choisi une unité adaptée au problème **ET** au système d'unités en vigueur
Explicitons cette notion sur le problème du coaxial.
On nous dit que le rayon intérieur du coaxial vaut r1 = 0,15 cm cela revient à dire que si l'on choisit comme unité de longueur le cm U(L) = 1 cm alors le rayon réduit vaut r1* = 0,15
Maintenant si l'on préfère choisir U(L) = r1 alors on trouve : r1* = 1 ; r2* = 0,50/0,15 =10/3 ; r3* = (r2* + e*) = 4
Nous choisirons naturellement comme unité de champ E, le champ maxi atteint en r1 soit U(E) = $\frac{Q}{2\,\pi\,L\,\varepsilon\,r1}$ (on a écrit L pour l pour éviter toute ambiguïté avec le chiffre 1 ...)

puis comme unité de potentiel U(V) = U(E) U(L) = $\frac{Q}{2\,\pi\,L\,\varepsilon}$

Il est donc toujours possible de revenir aux grandeurs SI en multipliant la mesure (grandeur réduite) par la valeur de l'unité.
Au prix de cet investissement initial, on gagne une grande simplification des formules manipulées ce qui les rend plus lisibles, un caractère universel des résultats (tous les câbles coaxiaux se ressemblent) et des graphes automatiquement bien dimensionnés.
Les raisonnements évidents souvent vus en cours et en exos ne sont pas repris.

```
> restart:
> U_E:=Q/(2*Pi*L*epsilon[o]*epsilon[r]*r1);epsilon[o]:=8.85e-12:epsilon[r]:=2:epsilon:=epsilon[o]*epsilon[r]:r1:=15e-4:L:=10:
```

$$U\_E := \frac{1}{2}\,\frac{Q}{\pi\,L\,\varepsilon_o\,\varepsilon_r\,r1}$$

```
> r2:=10/3:r3:=4: # nous sommes maintenant en grandeurs réduites
> E:=r->piecewise(r>1 and r<r2,1/r);
> plot(E,0..r3,title=`Champ E`,thickness=2);
```

$$E := r \rightarrow \mathrm{piecewise}\left(1 < r \text{ and } r < r2, \frac{1}{r}\right)$$



```
> V12:=int(E,1..r2);evalf(V12);#  § 1.4
```

$$V12 := \ln(2) + \ln(5) - \ln(3)$$

$$1.203972804$$

Pour exprimer la capacité par unité de longueur du câble, il faut déterminer l'unité réduite de capacité.

D'après $C = \frac{Q}{V_{12}}$, il est clair que si l'on choisit la charge Q du coaxial comme unité il est clair que $U(C) = \frac{1}{U(V)} = 2\,\pi\,L\,\varepsilon$ et que l'unité pour la capacité par unité de longueur vaut $U(CL) = 2\,\pi\,\varepsilon$

Cette valeur est assez satisfaisante puisque l'on sait que ε s'exprime justement en $\frac{F}{m}$

Finalement, la capacité réduite vaut simplement :

```
> C:=1/V12;evalf(C);U_CL:=2*Pi*epsilon;
```

$$C := \frac{1}{\ln(2) + \ln(5) - \ln(3)}$$

$$.8305835453$$

$$U\_CL := .3540\ 10^{-10}\ \pi$$

```
> C:=evalf(C*U_CL);# valeur SI en F/m
```

$$C := .9237117281\ 10^{-10}$$

On vient de trouver C/L = 92,4 pF / m. Calculons l'énergie (linéique) associée selon Wel = $\frac{C\,V_{12}^{\ 2}}{2}$,

Il est prudent ici de travailler en SI puisque l'on a déjà converti C en F et que l'on nous donne V12 en V

On en déduit les diverses unités

```
> We:= C*10^2/2*L;# on a multiplié par L la longueur du câble pour
  obtenir C total
```

$$We := .4618558641\ 10^{-7}$$

```
> U_V:=evalf(10/V12);
```

$$U\_V := 8.305835453$$

```
> U_E:=U_V/1.5e-4;#  Champ E(r1) en V/m
```

$$U\_E := 55372.23636$$

Champ B, énergie (et inductance) propre du câble

Le théorème d'Ampère nous dit que dans cette symétrie le champ magnétique est orthoradial fonction de r selon :

$B = \frac{\mu_o\, i(r)}{2\,\pi\,r}$ où i(r) est le courant enlacé dans un cercle de rayon r

Nous utiliserons comme unité de courant le courant total (I = 0,10 A)

On utilise toujours r1 comme unité de longueur

On utilise comme unité de champ B le champ B(r1) qui est d'ailleurs maximum

$U\_B = \frac{\mu_o I}{2 \pi r_1}$

Nous avons notés a , b , c les valeurs réduites des rayons r1, r2 , r3

I = 0,10 A $\mu_o = 4 \pi 10^{(-7)}$ H/m   L = 10 m (longueur totale du câble à ne pas confondre avec L1 sa "self")

Contrôler la pertinence des expressions proposées ci-dessous pour i(r)

```
> restart:
> i:=r->piecewise(r<a,r^2,r<b,1,r<c,(c^2-r^2)/(c^2-b^2)):'i(r)'=convert(i(r),piecewise,r);
```

$$\mathrm{i}(r) = \begin{cases} r^2 & r < a \\ 1 & r < b \\ \frac{c^2 - r^2}{c^2 - b^2} & r < c \end{cases}$$

```
> a:=1:b:=10/3:c:=4:plot(i,0..5,title=` i(r)`,thickness=2);
>
```




```
> B:=r->i(r)/r;B_:=convert(B(r),piecewise,r);plot(B,0..4,title=`B(r)`,numpoints=100,thickness=2);
```

$$B := r \to \frac{\mathrm{i}(r)}{r}$$

$$B\_ := \begin{cases} r & r \le 1 \\ \frac{1}{r} & r \le \frac{10}{3} \\ \frac{36}{11}\frac{1}{r} - \frac{9}{44} r & r \le 4 \\ 0 & 4 < r \end{cases}$$



Le champ B est continu puisque le courant i(r) l'est (absence de courant surfacique)

Pour obtenir le champ en Tesla il faut multiplier par l'unité de champ B soit :

```
> U_B:=mu[o]*It/(2*Pi*r1);mu[o]:=4e-7*Pi:It:=0.1:r1:=1.5e-4:#revalidez la ligne pour l'A.N.
```

$$U\_B := .0001333333334$$

Exprimons l'énergie magnétique U du câble à l'aide de la densité d'énergie.

$wm = \dfrac{B^2}{2\,\mu_o}$ en J/m^3 : densité volumique d'énergie

On peut toujours travailler en variables réduites.

Si l'on prend comme unité d'énergie la valeur :

$U\_Wm = \dfrac{\mu_o\, I_t^{\,2}\, L}{4\,\pi}$ valant simplement $U\_Wm = 10^{(-8)}$ J dans le problème PT 2008

On peut déterminer Wm(r) : énergie stockée dans le câble dans la région [0..r] avec r < c

Montrez qu'elle s'exprime comme une intégrale "assez simple" sur B^2 :

```
> Wm:=r->int(B(u)^2*u,u=0..r);plot(Wm,0..c,thickness=2);
```

$$Wm := r \rightarrow \int_0^r \mathrm{B}(u)^2\, u\, du$$



Déterminons l'influence relative des trois régions :

1 : r < a conducteur central

2 : a< r < b isolant

3 : b < r < c conducteur extérieur

La dernière expression (U_3) est la seule réellement laborieuse à établir (du moins à la main !)

Heureusement son calcul est rarement exigé aux concours...

```
> W1:=int(B(u)^2*u,u=0..1);W2:=int(B(u)^2*u,u=1..b);evalf(W2);W3:=int(B(u)^2*u,u=b..c);evalf(W3);
```

$$W1 := \frac{1}{4}$$

$$W2 := \ln(2) + \ln(5) - \ln(3)$$

$$1.203972804$$

$$W3 := -\frac{83}{44} + \frac{1296}{121}\ln(2) - \frac{1296}{121}\ln(5) + \frac{1296}{121}\ln(3)$$

$$.06643585$$

```
>
```

L'énergie associée à l'isolant (terme prépondérant) vaut W2 = 1,204 soit en SI W2 = 12,04. n J /m ou encore pour les 10 m de câble W2 =120,4. n J

On peut en déduire l'inductance propre puisque $W = \frac{L\,I_t^2}{2}$, il reste à savoir si l'on doit donner l'inductance par mètre ou l'inductance totale, l'examinateur oscillant entre ces deux notions...

L (total) = 24 µ H , L (par mètre) = 2,4 µ H / m

Résistance totale

Calculs très simples. Il suffit d'appliquer $R = \frac{L}{\gamma S}$ avec $S = \pi\, r_1^2$ pour le conducteur central et $S = \pi\,(r_3^2 - r_2^2)$ pour le conducteur extérieur

R1 = 2,9 mΩ / m ou encore R = 29 m Ω pour les 10 mètres de câble

Valeur négligeable devant la charge de 50 Ω , la tension Eg vaut donc sensiblement10 V ...